
Aufgabe 6.1

Die angegebene Routine zur Berechnung der n-ten Fibonacci-Zahl kann man direkt in Assembler übersetzen. Allerdings wäre eine iterative Lösung im Ernstfall vorzuziehen, da der Stackbedarf bei größeren Zahlen rapide anwächst und bereits berechnete Werte mehrfach neu berechnet werden.

Als Beispiel werden die 0.-, 1.-, 2.-, 5.- und 10.-Fibonacci-Zahl berechnet:

Datei fibo.mi:

    egyptTest: SEG

               MOVE W I H'10000', SP    -- Stack setzen

               MOVE W I 0, R0           -- Fibonacci-Zahl von 0 berechnen

               CALL fib                 -- Ergebnis liegt in R1

               MOVE W I 1, R0           -- Fibonacci-Zahl von 1 berechnen

               CALL fib                 -- Ergebnis in R1

               MOVE W I 2, R0           -- Fibonacci-Zahl von 2 berechnen

               CALL fib                 -- Ergebnis in R1

               MOVE W I 5, R0           -- Fibonacci-Zahl von 5 berechnen

               CALL fib                 -- Ergebnis in R1

               MOVE W I 10, R0          -- Fibonacci-Zahl von 5 berechnen

               CALL fib                 -- Ergebnis in R1

               HALT                     -- Programmende

   -- PROC fib : berechnet rekursiv die Fibonacci-Zahl der Eingabe

   -- in: R0 : Zahl n

   -- out: R1 : Fibonacci-Zahl von n

          fib: PUSHR                    -- Register auf Stack sichern

               CMP W I 0, R0

               JNE fib_test_1

               MOVE W I 0, R1

               JUMP fib_go_back

   fib_test_1: CMP W I 1, R0

               JNE fib_greater_1

               MOVE W I 1, R1

               JUMP fib_go_back

fib_greater_1:

               MOVE W R0, R2            -- R0 (n) in R2 zwischenspeichern

               SUB W I 1, R0            -- dec (n) -> n-1

               CALL fib

               MOVE W R1, R3            -- Rückgabe von fib(n-1) in 

-- R3 speichern

               SUB W I 1, R0            -- dec (n) -> n-2

               CALL fib

               ADD W R3, R1             -- R3 auf fib(n-2) aufaddieren 

-- und fertig...

  fib_go_back: MOVE W R1, 4+!SP         -- Rückgabe des Produkts in R1

-- (stack cheating)

               POPR                     -- Register zurückschreiben

               RET

   -- ENDP fib

               END

Aufgabe 6.2

Die Suchroutine durchläuft mit der binären Suche ein gegebenes, aufsteigend sortiertes Bytefeld rekursiv, bis der gesuchte Wert gefunden wurde oder bis alle Elemente ohne Erfolg durchsucht worden sind. Bei Fund liefert das Unterprogramm eine 1, bei Nicht-Fund eine 0 zurück. Als Beispiel werden 15 Werte nach zwei vorhandenen und einem nicht-enhaltenen Element durchsucht.

Datei suchen.mi:
    such_test: SEG

               MOVE W I H'10000', SP    -- Stack setzen

               JUMP go

field: DD B 1, 2, 10, 12, 14, 20, 25, 30, 32, 49, 51, 51, 52, 60,

 77

    field_len: DD W 15

           go:

               MOVE B I 60, R0          -- gesuchte Zahl = 60

               MOVEA field, R1          -- Startadresse

               MOVE W field_len, R2     -- Arraylänge

               CALL suchen              -- liefert 1, also gefunden

               MOVE B I 22, R0          -- gesuchte Zahl = 22

               MOVEA field, R1          -- Startadresse

               MOVE W field_len, R2     -- Arraylänge

               CALL suchen              -- liefert 0, also nicht gefunden

               MOVE B I 1, R0           -- gesuchte Zahl = 1

               MOVEA field, R1          -- Startadresse

               MOVE W field_len, R2     -- Arraylänge

               CALL suchen              -- liefert 1, also gefunden

               HALT                     -- Programmende

   -- PROC suchen : sucht rekursiv-binär die gegebene Zahl im

gegebenen, sortierten Feld

   --  in: R0 : gesuchte Zahl (Byte)

   --      R1 : Startadresse des sotierten Arrays (Wort)

   --      R2 : Anzahl der Einträge ( = Länge in Byte ) (Wort)

   -- out: R3 : 1 wenn gefunden, sonst 0 (Wort)

       suchen: PUSHR                    -- Register auf Stack sichern

               SUB W I 1, R2            -- Länge -1 = rechter Rand

               MOVE W R2, R3            -- rechten Rand setzen

               MOVE W I 0, R2             -- linken Rand setzen

               CALL rek_such            -- und ab in die Rekursion

               MOVE W R4, 12+!SP        -- Suchergebnis in R3 zurückgeben

  suchen_back: POPR                     -- Register zurückschreiben

               RET

   -- ENDP suchen

   -- PROC rek_such : der rekursive Part der Suchfunktion

   --  in: R0 : gesuchte Zahl (Byte)

   --      R1 : Startadresse des sortierten Abschnitts (Wort)

   --      R2 : linker Rand des Abschnitts (Wort)

   --      R3 : Rechter Rand des Abschnitts

   -- out: R4 : 1 wenn gefunden, sonst 0 (Wort)

     rek_such:

               MOVE W R0,-!SP

               MOVE W R1,-!SP

               MOVE W R2,-!SP

               MOVE W R3,-!SP

               CMP W R2, R3        -- Sofortige Austrittsbedingung prüfen

               JGT rek_not_found

               ADD W R2, R3, R4

               SH I -1, R4, R4

               CMP B R0, !R1/R4/

               JLT rek_links       -- wenn nicht matcht, dann wegspringen

               JGT rek_rechts

               MOVE W I 1, R4      -- Zahl gefunden!!!

               JUMP rek_such_back

rek_links:     MOVE W R4, R3       -- rechte Grenze neu setzen -> linke

-- Hälfte weitersuchen

               SUB W I 1, R3       -- einen weniger am rechten Rand, 

-- den alten hatte man ja gerade

               CALL rek_such

               JUMP rek_such_back

rek_rechts:    MOVE W R4, R2       -- linke Grenzen neu setzen -> 

-- rechte Hälfte weitersuchen

               ADD W I 1, R2       -- einen mehr am linken Rand, den 

-- alten hatte man ja gerade

               CALL rek_such

               JUMP rek_such_back

rek_not_found:

               MOVE W I 0, R4      -- Zahl nicht gefunden

rek_such_back:

               MOVE W !SP+, R3

               MOVE W !SP+, R2

               MOVE W !SP+, R1

               MOVE W !SP+, R0

               RET

   -- ENDP rek_such

               END
Aufgabe 6.3

Leider ist die Header-Version des Arrays auf den Vorlesungs-Folien nicht sonderlich präzise formuliert. Da die MI mit einem 32-Bit-Adressraum arbeitet, sind für Speicherung Adressen auch 32 Bit - Felder nötig. Das erste Feld muß daher - unabhängig von der Datenbreite - 32 bit lang sein. Ebenso wäre es ratsam, die folgenden Felder mit der Spannweite der Dimensionen mit 32 - bit - Feldern zu auszustatten, um nach oben unbeschränkt zu sein.

Klar ist uns auch nicht, wie man das Ende der ersten Dimension erreichen soll, da S1 nicht im Header enthalten ist. Notfalls könnte man diesen Wert über die Größe des Gesamtarrays berechnen, aber diese Angabe soll ja gerade nicht übergeben werden.

Ferner fehlen jetzt auch die notwendigen Startwerte für alle Dimensionen. Durch die adr0-Berechnung kann man zwar schön mit den angegebenen Wertebereichen arbeiten, fängt man allerdings bei 0 zu zählen an, landet man schnell in "negativen Speicherbereichen". Es muß also bei mi angefangen werden. Diese Werte sind allerdings innerhalb des Unterprogramms unbekannt, da sie nicht im Array-Header abgelegt werden.

Alternativ schlagen wir daher vor, die Spannweiten jeder Dimension im Header in 32-Bit-Feldern abzulegen und adr0 und die Adressen eines beliebigen Elements folgendermaßen zu berechnen:

Wobei basis die Basismaschinenadresse der Struktur im Speicher, k die Anzahl der Dimensionen, si die Spannweite und mi die linke Grenze der Wertebereiche sind.

Für ein Feld mit den Abmessungen [ -5:3, 12:17, 1:31 ] ergibt sich dann:

s1 = 9

s2 = 6

s3 = 31

Und dementsprechend 

adr0 = basis - 153

Auf die Erweiterung des Assemblerprogrammes haben wir aus Zeitgründen verzichtet. Allerdings wäre eine Implementierung relativ einfach, da die eigentliche binäre Suche schon in der einfachen Version in einem eigenen Unterprogramm codiert ist. Man müßte jetzt lediglich die Hauptprozedur (suchen) um das Durchlaufen des Multidimensionalen Arrays erweitern. Dazu würde man einfach ab der zweiten Dimension alle Kombinationen durchlaufen und jeweils mit diesen Koordinaten den ersten Block mit dem jetzigen rek_such-Unterprogramm durchsuchen (so käme man auf s2*s3*...*sk Durchläufe bzw. Blöcke).

